

APPENDIX B

Datalogger Program for Campbell 21X

Campbell Program for Thermostat Testing Study

(enter an "A" after each instruction)

INSTRUCTION	DESCRIPTION
*1	Enter Table 1
5	5 sec. Execution Interval
P17	TC Reference temp.
1	location
P14	TC - differential ;check thermocouple
1	reps
11	5 mv, 250 usec
1	in chan
1	type 1
1	ref. location
4	location
.018	multiplier
.32	offset
P2	VOLTS - differential ;check relative humidity
1	reps
15	5000 mv, 250 usec
2	in chan
5	location
1	multiplier
0	offset
P2	VOLTS - differential ;check thermostat voltage
1	reps
15	5000 mv, 250 usec
3	in chan
2	location
1	multiplier
0	offset
P33	Z = X + Y ;sum TC readings
4	location of X
6	location of Y
6	location of Z
P33	Z = X + Y ;sum RH readings
5	location of X
7	location of Y
7	location of Z
P32	Z = Z + 1 ;increment counter
3	location
P89	IF voltage is less than F ;check if unit is on
2	location
4	X .lt. F
25	F
30	THEN DO (needs endif)
P91	IF flag
21	if flag 1 is low
30	THEN DO (needs endif)
P38	Z = X / Y
6	location X
3	location Y
6	location Z
P37	Z = X * F
4	location X
100	F
4	location Z
P37	Z = X * F
6	location X
100	F
6	location Z
P38	Z = X / Y
7	location X
3	location Y
7	location Z
P30	Z = F
1	F
8	location Z
P91	IF flag ;set flag 0 high and move to final storage
20	do if flag 0 is low
10	set flag 0 high (doesn't need endif)
P77	TIME
1111	YY:DD:HH:MM:SS
P70	SAMPLE
5	reps
4	start location

```

P89      IF X.lt.F
2        location
4        X.lt.F
30       F
11       set flag 1 high
P30      Z = F ;reset totals
0        F
3        location
P30      Z = F
0        F
6        location
P30      Z = F
0        F
7        location
P95      END
P95      END

P89      IF voltage is greater than F ;check if unit is off
2        location
3        X.ge.F
25       F
30       THEN DO (needs endif)
P91      IF flag
11       if flag 1 is high
30       THEN DO (needs endif)
P38      Z = X / Y
6        location X
3        location Y
6        location Z
P37      Z = X * F
4        location X
100      F
4        location Z
P37      Z = X * F
6        location X
100      F
6        location Z
P38      Z = X / Y
7        location X
3        location Y
7        location Z
P30      Z = F
0        F
8        location Z

P91      IF flag ;set flag 0 high and move to final storage
20       do if flag 0 is low
10       set flag 0 high (doesn't need endif)
P77      TIME
1111     YY:DD:HH:MM
P70      SAMPLE
5        reps
4        start location
P89      IF X.ge.F
2        location
3        X.ge.F
30       F
21       set flag 1 low
P30      Z = F ;reset totals
0        F
3        location
P30      Z = F
0        F
6        location
P30      Z = F
0        F
7        location
P95      END
P95      END

```

Campbell Channel
(front panel)
1 TC
2 RH
3 Istat volts

Intermediate Storage
(*6 mode)
1 Panel Temp
2 Istat volts
3 # of meas.
4 TC
5 RH
6 Tot. TC
7 Tot. RH
8 Counts

Final Storage
(*7 mode)
1 21x ID
2 Year
3 Julian Day
4 HH:MM:
5 SS
6 TC
7 RH
8 Avg. TC
9 Avg. RH
10 Flag *

* Flag - 0 = unit turned off this scan
- 1 = unit turned on this scan

Campbell Programming Guide

- Entering data.....two tables are available (1 and 2) with a third table available for subroutines. Enter *1 for table 1.
- Compiling.....after program has been entered, compile with *0, *6, *B, and *D.
 *0, *B, *D.....compiles; output ports and flags are set low, the timer is reset, and data values in input and intermediate storage are RESET TO ZERO.
 *6.....compiles; ports, flags, timer, and data are UNALTERED.
- Setting time.....*5A; enter year (A), julian day (A), hour-min (A), and sec (A).
- Internal memory...*A mode; displays memory allocation.
 To reset FINAL storage without altering program, enter in same value and recompile with *6, then *0 (the compile function is only executed after a program change has been made (i.e. *6 will compile then *0 won't erase input or intermediate storage)).
- Final Storage.....*7 mode; use A (advance) or B (back-up) to Data will be in final storage only when unit turns on or off. check data.
- Intermediate St... *6 mode; shows the data that is collected by the program. Only selected portions will be transferred to Final Storage.

Star (*) Mode Summary

Key	Mode
*0	LOG data and indicate active tables
*1	Program Table 1
*2	Program Table 2
*3	Program Table 3
*4	Enable/disable tape and/or printer output
*5	Display/set real time clock
*6	Display/alter Input Storage data, toggle flags
*7	Display Final Storage data
*8	Final Storage data transfer to cassette tape
*9	Final Storage data transfer to printer
*A	Memory allocation/reset
*B	Signature test
*C	Security
*D	Save/load program

Key Description/Editing Functions

Key	Action
0-9	Key numeric entries into display
*	Enter Mode (followed by Mode Number)
A	Enter/advance
B	Back up
C	Change the sign of a number or index an Input Location to loop counter
D	Enter the decimal point
#	Clear the right most digit keyed into the display
#A	Advance to the next instruction in program table
#B	(*1, *2, *3) or to next Output Array in Final Storage(*7)
#B	Back up to previous instruction in program table or to previous Output Array in Final Storage
#D	Delete entire instruction